

# **Project**

## **Class-Based Verification for a Synchronous FIFO USING UVM**

By: Mohamed Sayed Mohamed Soliman

# Contents

1. Introduction to the Synchronous FIFO.....	3
2. UVM testbench showing the UVM structure.....	5
3. Design RTL code .....	8
4. verification plan.....	10
5. Table to the assertion.....	10
6. Verification code	
1. 6.1. top module .....	11
2. 6.2. interface .....	11
3. 6.3. FIFO test .....	12
4. 6.4. FIFO env.....	13
5. 6.5. FIFO reset sequence.....	14
6. 6.6 . FIFO main sequence.....	14
7. 6.7. FIFO scoreboard .....	16
8. 6.8. FIFO coverage .....	18
9. 6.9. FIFO Agent.....	20
10.6.10. Driver.....	21
11.6.11. monitor.....	22
12.6.12. FIFO sequence item.....	23
13.6.13. FIFO sequencer.....	24
7. do file.....	24
8. Transcript.....	24
9. Waveform write sequence.....	25
10. Waveform read sequence.....	25
11. Waveform write and read sequence.....	25
12. Assertion.....	26
13. Assertion coverage.....	26
14. Code Coverage.....	26
15. Functional Coverage report.....	28

# 1. Introduction to the Synchronous FIFO

**A Synchronous FIFO (First-In-First-Out) :** is a fundamental hardware component used in digital systems for data buffering and transferring between different subsystems. It ensures that data is read in the same order as it was written, and both write and read operations are synchronized to a common clock. This makes it particularly useful for applications where data flow needs to be managed in a controlled and sequential manner, often between two devices or subsystems operating at different speeds but synchronized to the same clock.

## Key Features of a Synchronous FIFO:

1. **Single Clock Domain:** Both read and write operations are synchronized to the same clock signal, eliminating the need for complex clock domain crossing techniques.
2. **Data Storage:** The FIFO stores data in an internal memory (typically implemented as a register array or RAM). The depth (number of entries) and width (bits per entry) of the FIFO can be parameterized to fit the specific requirements of the application.
3. **Control Flags:**
  - **Full:** Indicates the FIFO is full and cannot accept more write operations until space is freed by reading.
  - **Empty:** Indicates the FIFO is empty and no data is available for reading.
  - **Almost Full/Almost Empty:** Optional flags that provide an early warning before the FIFO becomes completely full or empty.
4. **Read/Write Pointers:** The FIFO internally maintains pointers for tracking the current positions for reading and writing, ensuring data is managed in a circular buffer fashion.

## Applications of Synchronous FIFO:

- **Data Rate Matching:** Used between subsystems with different data rates but synchronized clocks (e.g., between a CPU and a memory controller).
- **Buffering:** Temporary storage for bursty data streams or packet data, smoothing out fluctuations in data rates.
- **Pipeline Processing:** Helps in decoupling stages of a digital pipeline, allowing one stage to write data while the next stage reads it at a different rate.

## Operational Overview:

- **Write Operation:** Data is written to the FIFO at the location pointed to by the write pointer. After a successful write, the write pointer increments, and the **full** flag is updated if necessary.
- **Read Operation:** Data is read from the FIFO at the location pointed to by the read pointer. The read pointer increments after a read operation, and the **empty** flag is updated accordingly.


The FIFO must manage these operations without data loss or overwriting, maintaining data integrity across various conditions.

## Design Considerations:

- **Depth:** The size of the FIFO (number of entries) is chosen based on the maximum burst length or data buffering requirements of the system.
- **Width:** The data width is parameterized, depending on the size of the data being transferred (e.g., 8-bit, 16-bit, or 32-bit).
- **Latency:** The delay between a write or read request and when the data becomes available is minimal in a well-designed FIFO.

Synchronous FIFOs are widely used in digital designs, including networking hardware, processors, and communication systems, where reliable and ordered data transfer is critical.

## 2. UVM testbench showing the UVM structure



In the UVM testbench you've shared, the structure of the verification environment for a FIFO (First-In-First-Out) design follows a modular and hierarchical approach. I'll now describe how this UVM testbench works, based on your diagram, starting from the top module to driving the interface, monitoring the outputs, and analyzing the results.

---

### 1. Top Module (FIFO\_TOP and Test Instantiation)

At the highest level, the **FIFO\_TOP** module includes:

- **DUT** (Design Under Test): The FIFO module you're verifying.
- **Interface**: The interface connects the DUT to the testbench. This interface contains all the signal connections like the inputs (data, write\_enable, read\_enable) and outputs (data\_out, full, empty).

The interface simplifies passing signals between the DUT and the verification environment.

## 2. Driving the Interface (FIFO Driver and Sequence)

In this verification flow, input stimulus is generated by the UVM sequence and is driven onto the DUT through the driver.

- **FIFO\_seq\_item**: This represents the individual transaction (or data item) that is applied to the DUT. For example, a write or read operation could be a transaction in the FIFO sequence item.
- **FIFO\_sequence**: This generates sequences of these FIFO\_seq\_items. The sequence could contain various write and read operations or random patterns to test the FIFO's functionality.
- **FIFO\_driver**: The driver receives transactions from the sequencer and converts them into low-level signals that drive the interface of the DUT. The driver ensures that the correct signals (e.g., data\_in, write\_enable, read\_enable) are driven onto the interface pins in synchronization with the clock.
- **Sequencer**: The **FIFO\_sequencer** manages the flow of the sequence. It picks the sequence items from the FIFO\_sequence and forwards them to the driver. This ensures that the input stimulus is applied in the correct order and at the appropriate times.

## 3. Monitoring (FIFO Monitor and Virtual Interface)

- **FIFO\_monitor**: This component passively monitors the signals on the interface. It captures both the input transactions (e.g., write commands) and output responses (e.g., data read from the FIFO). The monitor converts these signal-level interactions back into transaction-level representations.
- The monitor does not interfere with the signal flow but extracts the necessary information to ensure that the DUT is behaving as expected.
- The monitor also works with the **virtual interface**, which provides the necessary connections to observe the signal activities without physically modifying the interface.

## 4. Analyzing the Output (Scoreboard and Coverage)

- **FIFO\_scoreboard**: This is where the actual checking occurs. The scoreboard compares the monitored transactions against expected results. For example, if a write operation is performed, the scoreboard ensures that the data is stored in the FIFO and can be

read out in the correct order. If any mismatches occur (such as reading incorrect data), the scoreboard raises an error.

- **FIFO\_coverage:** This component tracks how much of the design has been exercised during the test. Coverage ensures that all functional aspects of the FIFO (like full, empty, overflow, and underflow conditions) have been tested. It records the various operations (write, read, overflow, etc.) and checks whether all scenarios have been covered.

### **5. Overall Test Flow**

- The test starts at the **FIFO\_TEST** level, where the testbench environment, including the scoreboard, monitor, driver, and coverage, is instantiated.
- The sequencer triggers the sequences, which generate the transactions for the driver to send to the DUT.
- The monitor observes the interactions and forwards them to the scoreboard for checking.
- Coverage collects statistics on which parts of the design have been exercised.

### **Conclusion**

This UVM testbench follows a structured approach to verifying the FIFO design by modularizing the components for stimulus generation, driving the interface, monitoring signals, and analyzing the output. The modular structure makes the testbench reusable, configurable, and scalable for different scenarios and complexities in design verification.

### 3. Design RTL code

```
7 ///////////////////////////////////////////////////////////////////
8 module FIFO(FIFO_if.DUT  FIFOif);
9
10 localparam max_fifo_addr = $clog2(FIFOif.FIFO_DEPTH);
11
12 reg [FIFOif.FIFO_WIDTH-1:0] mem [FIFOif.FIFO_DEPTH-1:0];
13
14 reg [max_fifo_addr-1:0] wr_ptr, rd_ptr;
15 reg [max_fifo_addr:0] count;
16
17 always @(posedge FIFOif.clk or negedge FIFOif.rst_n) begin
18     if (!FIFOif.rst_n) begin
19         wr_ptr <= 0;
20     end
21     else if (FIFOif.wr_en && count < FIFOif.FIFO_DEPTH) begin
22         mem[wr_ptr] <= FIFOif.data_in;
23         FIFOif.wr_ack <= 1;
24         wr_ptr <= wr_ptr + 1;
25     end
26     else begin
27         FIFOif.wr_ack <= 0;
28         if (FIFOif.full && FIFOif.wr_en)
29             FIFOif.overflow <= 1;
30         else
31             FIFOif.overflow <= 0 ;
32     end
33 end
```



```

35  always @(posedge FIFOif.clk or negedge FIFOif.rst_n) begin
36      if (!FIFOif.rst_n) begin
37          rd_ptr <= 0;
38          FIFOif.data_out <= 0;
39      end
40      else if (FIFOif.rd_en && count != 0) begin
41          FIFOif.data_out <= mem[rd_ptr];
42          rd_ptr <= rd_ptr + 1;
43      end
44      else begin
45          if (FIFOif.empty && FIFOif.rd_en)
46              FIFOif.underflow <= 1;
47          else
48              FIFOif.underflow <= 0 ;
49      end
50  end

51
52  always @(posedge FIFOif.clk or negedge FIFOif.rst_n) begin
53      if (!FIFOif.rst_n) begin
54          count <= 0;
55      end
56      else begin
57          if ( ({FIFOif.wr_en, FIFOif.rd_en} == 2'b10) && !FIFOif.full)
58              count <= count + 1;
59          else if ( ({FIFOif.wr_en, FIFOif.rd_en} == 2'b01) && !FIFOif.empty)
60              count <= count - 1;
61          else if ( ({FIFOif.wr_en, FIFOif.rd_en} == 2'b11) && FIFOif.empty)
62              count <= count + 1;
63          else if ( ({FIFOif.wr_en, FIFOif.rd_en} == 2'b11) && FIFOif.full)
64              count <= count - 1;
65      end
66  end

67
68  assign FIFOif.full = (count == FIFOif.FIFO_DEPTH)? 1 : 0;
69  assign FIFOif.empty = (count == 0)? 1 : 0;
70  assign FIFOif.almostfull = (count == FIFOif.FIFO_DEPTH-1)? 1 : 0;
71  assign FIFOif.almostempty = (count == 1)? 1 : 0;
72
73
74  endmodule

```

## 4. Verification plan

Label	Description	Stimulus Generation	Functional Coverage	Functionality Check
FIFO_1	When the reset is asserted, the output value should be low & All the location of FIFO is cleared	Directed at the start of the sim, then randomized with constraint that drive the reset to be off most of the simulation time.	-	A check_data function in the FIFO_scoreboard to make sure the output is correct
FIFO_2	The first sequence when the write enable is asserted and the read enable is deasserted for 100 iteration ==> we will write data_in inside the FIFO	Randomization under constraints on the data_in & write enable	include cover point for the wr_en signal & all flag signal	A check_data function in the FIFO_scoreboard to make sure the output is correct
FIFO_3	The second sequence When the write enable is deasserted and the read enable is asserted for 100 iteration ==> we will read from the FIFO & get the value to be stored in data_out signal	Randomization under constraints for read enable signal	include cover point for the rd_en signal & all flag signal	A check_data function in the FIFO_scoreboard to make sure the output is correct
FIFO_4	The last sequence when the write enable is asserted and the read enable is asserted for 10000 iteration	Randomization under constraints on the data_in , write enable and read enable	-	A check_data function in the FIFO_scoreboard to make sure the output is correct

## 5. Table to the assertion

Feature	Assertion
<b>Whenever the FIFO is not full and wr_en is high, wr_ack is high</b>	@(posedge clk) (wr_en && count < FIFO_DEPTH)   => wr_ack
<b>Whenever the FIFO is full and wr_en is high, overflow signal is asserted</b>	@(posedge FIFOif.clk) (FIFOif.full && FIFOif.wr_en)   => overflow
<b>When count == 0, FIFO is empty, not full, almostempty and almostfull signals are also cleared</b>	assert (empty && !full && !almostempty && !almostfull)
<b>When count == 1, FIFO is not empty, not full, almost empty is set, and almost full is clear</b>	assert (!empty && !full && almostempty && !almostfull)
<b>When count == FIFO_DEPTH - 1, FIFO is not empty or full, but almost full is asserted</b>	assert (!empty && !full && !almostempty && almostfull)
<b>When count == FIFO_DEPTH, FIFO is full, not empty, and almost signals are clear</b>	assert (!empty && full && !almostempty && !almostfull)
<b>Whenever FIFO is empty and rd_en is high, underflow signal is asserted</b>	@(posedge clk) (empty && rd_en)   => underflow
<b>When wr_en is high , rd_en is low and FIFO is not full , count increments by 1</b>	@(posedge clk)(wr_en&&!rd_en&&!full)   =>( count === (\$past(count) + 1 )
<b>When rd_en is high ,wr_en is low and FIFO is not empty,count decrements by 1</b>	@(posedge clk)(!wr_en&&rd_en&&!empty)   =>( count === (\$past(count) - 1 )
<b>When both wr_en and rd_en are high and FIFO is empty, count increments by 1</b>	@(posedge clk)(wr_en&&rd_en&&empty)   =>( count === (\$past(count) + 1 )
<b>When both wr_en and rd_en are high and FIFO is full, count decrements by 1</b>	@(posedge clk)(wr_en&&rd_en&&full)   =>( count === (\$past(count) - 1 )

## 6. Verification code

### 6.1. top module

```
1  import FIFO_test_pkg::*;
2  import uvm_pkg::*;
3  `include "uvm_macros.svh"
4
5
6
7  module FIFO_top();
8  bit clk ;
9
10 initial
11 begin
12     clk = 0 ;
13     forever
14     begin
15         #1 clk = ~clk ;
16     end
17 end
18 FIFO_if FIFOif (clk) ;
19 FIFO du (FIFOif) ;
20 bind FIFO SVA sva(FIFOif) ;
21 initial
22 begin
23     uvm_config_db#(virtual FIFO_if)::set(null , "uvm_test_top" , "FIFOif" , FIFOif );
24     run_test("FIFO_test");
25 end
26
27
28
29
30 endmodule
```

### 6.2. interface

```
1  interface FIFO_if(clk) ;
2
3      parameter FIFO_WIDTH = 16;
4      parameter FIFO_DEPTH = 8;
5      localparam max_fifo_addr = $clog2(FIFO_DEPTH);
6
7      input clk;
8      logic [FIFO_WIDTH-1:0] data_in;
9      logic rst_n, wr_en, rd_en;
10     logic [FIFO_WIDTH-1:0] data_out;
11     logic wr_ack, overflow;
12     logic full, empty, almostfull, almostempty, underflow;
13
14
15     modport DUT (input clk , rst_n, wr_en, rd_en , data_in , output data_out , wr_ack, overflow , full, empty, almostfull, almostempty, underflow );
16
17
18 endinterface
```

### 6.3. FIFO test

```
1 package FIFO_test_pkg ;
2
3 import FIFO_env_pkg::*;
4 import FIFO_config_pkg::*;
5 import FIFO_seq_reset_pkg::*;
6 import FIFO_seq_main_pkg::*;
7 import uvm_pkg::* ;
8 `include "uvm_macros.svh"
9
10
11 class FIFO_test extends uvm_test;
12 | `uvm_component_utils(FIFO_test)
13 | FIFO_env env ;
14 | FIFO_config FIFO_config_obj_test ;
15 | FIFO_reset_sequence reset_sequence ;
16 | FIFO_write_sequence write_sequence ;
17 | FIFO_read_sequence read_sequence ;
18 | FIFO_read_and_write_sequence read_and_write_sequence ;
19
20 | function new(string name = "FIFO_test" , uvm_component parent = null );
21 | | super.new(name ,parent ) ;
22 | endfunction
23
24 function void build_phase(uvm_phase phase) ;
25 | super.build_phase(phase);
26 | env = FIFO_env::type_id::create("env",this );
27 | FIFO_config_obj_test = FIFO_config::type_id::create("FIFO_config_obj_test");
28 | reset_sequence = FIFO_reset_sequence::type_id::create("reset_sequence");
29 | write_sequence = FIFO_write_sequence::type_id::create("write_sequence");
30 | read_sequence = FIFO_read_sequence::type_id::create("read_sequence");
31 | read_and_write_sequence = FIFO_read_and_write_sequence::type_id::create("read_and_write_sequence");
32 | if(!uvm_config_db#(virtual FIFO_if)::get (this , "" , "FIFOif" , FIFO_config_obj_test.FIFO_vif ))
33 | | `uvm_fatal("run_phase" , "test - unable to get the virtual interface ") ;
34
35 | uvm_config_db#(FIFO_config)::set (this , "*" , "CFG" , FIFO_config_obj_test ) ;
36
37 endfunction
```

```

38
39 task run_phase(uvm_phase phase) ;
40     super.run_phase(phase);
41     phase.raise_objection(this);
42     `uvm_info("run_phase","reset assert" , UVM_LOW)
43     reset_sequence.start(env.agt.sqr);
44     `uvm_info("run_phase","reset deassert" , UVM_LOW)
45
46     `uvm_info("run_phase","write stimulus generation started" , UVM_LOW)
47     write_sequence.start(env.agt.sqr);
48     `uvm_info("run_phase","write stimulus generation ended" , UVM_LOW)
49
50     `uvm_info("run_phase","read stimulus generation started" , UVM_LOW)
51     read_sequence.start(env.agt.sqr);
52     `uvm_info("run_phase","read stimulus generation ended" , UVM_LOW)
53
54
55     `uvm_info("run_phase","read and write stimulus generation started" , UVM_LOW)
56     read_and_write_sequence.start(env.agt.sqr);
57     `uvm_info("run_phase","read and write stimulus generation ended" , UVM_LOW)
58     phase.drop_objection(this);
59     endtask:run_phase
60
61 endclass
62 endpackage

```

#### 6.4. FIFO env

```

1 package FIFO_env_pkg ;
2
3 import FIFO_agent_pkg::*;
4 import FIFO_scoreboard_pkg::*;
5 import FIFO_coverage_pkg::*;
6 import uvm_pkg::* ;
7 `include "uvm_macros.svh"
8
9
10 class FIFO_env extends uvm_env;
11     `uvm_component_utils(FIFO_env)
12
13     FIFO_agent agt ;
14     FIFO_scoreboard sb ;
15     FIFO_coverage cov ;
16     function new(string name = "FIFO_env" , uvm_component parent = null );
17     |     super.new(name ,parent ) ;
18     endfunction
19
20     function void build_phase(uvm_phase phase) ;
21     |     super.build_phase(phase);
22     |     agt = FIFO_agent::type_id::create("agt",this );
23     |     sb = FIFO_scoreboard::type_id::create("sb",this );
24     |     cov = FIFO_coverage::type_id::create("cov",this );
25     endfunction
26
27     function void connect_phase(uvm_phase phase) ;
28     |     super.connect_phase(phase);
29     |     agt.agt_ap.connect(sb.sb_export);
30     |     agt.agt_ap.connect(cov.cov_export);
31     endfunction
32 endclass
33 endpackage

```

## 6.5. FIFO reset sequence

```
1 package FIFO_seq_reset_pkg ;
2
3 import FIFO_seq_item_pkg::*;
4 import uvm_pkg::* ;
5 `include "uvm_macros.svh"
6
7
8 class FIFO_reset_sequence extends uvm_sequence #(FIFO_seq_item);
9     `uvm_object_utils(FIFO_reset_sequence)
10
11     FIFO_seq_item seq_item ;
12     function new(string name = "FIFO_reset_sequence" );
13         super.new(name) ;
14     endfunction
15     task body;
16         seq_item = FIFO_seq_item::type_id::create("seq_item");
17         start_item(seq_item);
18         seq_item.rst_n = 0 ;
19         seq_item.data_in = 0 ;
20         seq_item.wr_en = 0 ;
21         seq_item.rd_en = 0 ;
22         finish_item(seq_item);
23
24     endtask
25 endclass
26 endpackage
```

## 6.6. FIFO main sequence

```
1 package FIFO_seq_main_pkg;
2
3 import FIFO_seq_item_pkg::*;
4 import uvm_pkg::* ;
5 `include "uvm_macros.svh"
6
7
8 class FIFO_write_sequence extends uvm_sequence #(FIFO_seq_item);
9     `uvm_object_utils(FIFO_write_sequence)
10
11     FIFO_seq_item seq_item ;
12     function new(string name = "FIFO_write_sequence" );
13         super.new(name) ;
14     endfunction
15     task body;
16         repeat(100)
17         begin
18             seq_item = FIFO_seq_item::type_id::create("seq_item");
19             start_item(seq_item);
20             seq_item.constraint_mode(0);
21             seq_item.write_only.constraint_mode(1);
22             assert(seq_item.randomize()) ;
23             finish_item(seq_item);
24         end
25
26     endtask
27 endclass
```

```

30  class FIFO_read_sequence extends uvm_sequence #(FIFO_seq_item);
31      `uvm_object_utils(FIFO_read_sequence)
32
33      FIFO_seq_item seq_item ;
34      function new(string name = "FIFO_read_sequence" );
35          super.new(name) ;
36      endfunction
37      task body;
38          repeat(100)
39              begin
40                  seq_item = FIFO_seq_item::type_id::create("seq_item");
41                  start_item(seq_item);
42                  seq_item.constraint_mode(0);
43                  seq_item.read_only.constraint_mode(1);
44                  assert(seq_item.randomize()) ;
45                  finish_item(seq_item);
46              end
47          endtask
48      endclass
49

```

```

51
52  class FIFO_read_and_write_sequence extends uvm_sequence #(FIFO_seq_item);
53      `uvm_object_utils(FIFO_read_and_write_sequence)
54
55      FIFO_seq_item seq_item ;
56      function new(string name = "FIFO_read_and_write_sequence" );
57          super.new(name) ;
58      endfunction
59      task body;
60          repeat(10000)
61              begin
62                  seq_item = FIFO_seq_item::type_id::create("seq_item");
63                  start_item(seq_item);
64                  seq_item.constraint_mode(0);
65                  seq_item.read_and_write.constraint_mode(1);
66                  assert(seq_item.randomize()) ;
67                  finish_item(seq_item);
68              end
69          endtask
70      endclass
71  endpackage
72

```

## 6.7. FIFO scoreboard

```
1  package FIFO_scoreboard_pkg ;
2
3  import FIFO_seq_item_pkg::* ;
4  import uvm_pkg::* ;
5  `include "uvm_macros.svh"
6
7
8  class FIFO_scoreboard extends uvm_scoreboard;
9      `uvm_component_utils(FIFO_scoreboard)
10
11      uvm_analysis_export #(FIFO_seq_item) sb_export ;
12      uvm_tlm_analysis_fifo #(FIFO_seq_item) sb_fifo ;
13      FIFO_seq_item seq_item_sb ;
14
15      parameter FIFO_WIDTH = 16 ;
16      parameter FIFO_DEPTH = 8 ;
17      localparam max_fifo_addr = $clog2(FIFO_DEPTH);
18      logic [FIFO_WIDTH-1:0] data_out_ref = 0;
19      logic wr_ack_ref = 0 ;
20      logic overflow_ref = 0 ;
21      logic full_ref = 0 ;
22      logic empty_ref = 0 ;
23      logic almostfull_ref = 0 ;
24      logic almostempty_ref = 0 ;
25      logic underflow_ref = 0 ;
26      logic [6:0] flag_test , flag_dut ;
27      logic [FIFO_WIDTH-1:0] mem_queue[$] ;
28      logic [max_fifo_addr:0] count = 0 ;
29
30      int error_count = 0 ;
31      int correct_count = 0 ;
32      function new(string name = "FIFO_scoreboard" , uvm_component parent = null );
33          super.new(name ,parent ) ;
34      endfunction
35
36      function void build_phase(uvm_phase phase) ;
37          super.build_phase(phase);
38          sb_export = new("sb_export" , this) ;
39          sb_fifo = new("sb_fifo" , this) ;
40      endfunction
41
42      function void connect_phase(uvm_phase phase) ;
43          sb_export.connect(sb_fifo.analysis_export);
44      endfunction
45
```



```

46
47 task run_phase(uvm_phase phase) ;
48     super.run_phase(phase);
49     forever
50     begin
51         sb_fifo.get(seq_item_sb) ;
52         ref_model(seq_item_sb) ;
53         flag_test = { wr_ack_ref , overflow_ref , full_ref , empty_ref , almostfull_ref , almostempty_ref , underflow_ref } ;
54         flag_dut = { seq_item_sb.wr_ack , seq_item_sb.overflow , seq_item_sb.full , seq_item_sb.empty , seq_item_sb.almostfull \
55             , seq_item_sb.almostempty , seq_item_sb.underflow } ;
56         if((seq_item_sb.data_out != data_out_ref) || (flag_dut != flag_test))
57         begin
58             `uvm_error("run_phase" , $sformatf("comparsion failed trasnsaction received by the dut %s shile the reference out %ob" \
59                 ,seq_item_sb.convert2string , data_out_ref )) ;
60             error_count++ ;
61         end
62         else
63         begin
64             `uvm_info("run_phase" ,seq_item_sb.convert2string() , UVM_HIGH ) ;
65             correct_count++ ;
66         end
67     end
68 endtask

```

```

67
68 task ref_model(FIFO_seq_item seq_item);
69     fork
70     begin
71         if (!seq_item.rst_n) begin
72             mem_queue.delete();
73         end
74         else if (seq_item.wr_en && count < seq_item.FIFO_DEPTH) begin
75             mem_queue.push_back(seq_item.data_in) ;
76             wr_ack_ref = 1;
77         end
78         else begin
79             wr_ack_ref = 0;
80             if (full_ref & seq_item.wr_en)
81                 overflow_ref = 1;
82             else
83                 overflow_ref = 0;
84         end
85     end
86     begin
87         if (!seq_item.rst_n) begin
88             data_out_ref = 0 ;
89         end
90         else if (seq_item.rd_en && count != 0) begin
91             data_out_ref = mem_queue.pop_front() ;
92         end
93         else begin
94             if (empty_ref && seq_item.rd_en)
95                 underflow_ref = 1;
96             else
97                 underflow_ref = 0 ;
98         end
99     end

```

```

104     if (!seq_item.rst_n) begin
105         count = 0;
106     end
107     else begin
108         if ( ({seq_item.wr_en, seq_item.rd_en} == 2'b10) && !full_ref)
109             count = count + 1;
110         else if ( ({seq_item.wr_en, seq_item.rd_en} == 2'b01) && !empty_ref)
111             count = count - 1;
112         else if ( ({seq_item.wr_en, seq_item.rd_en} == 2'b11) && empty_ref)
113             count = count + 1;
114         else if ( ({seq_item.wr_en, seq_item.rd_en} == 2'b11) && full_ref)
115             count = count - 1;
116     end
117     full_ref = (count == FIFO_DEPTH)? 1 : 0;
118     empty_ref = (count == 0)? 1 : 0;
119     almostfull_ref = (count == FIFO_DEPTH-1)? 1 : 0;
120     almostempty_ref = (count == 1)? 1 : 0;
121
122     endtask
123
124     function void report_phase(uvm_phase phase) ;
125     super.report_phase(phase);
126     `uvm_info("report_phase", $sformatf("total successful %0d ", correct_count ), UVM_MEDIUM ) ;
127     `uvm_info("report_phase", $sformatf("total FAILED %0d ", error_count ), UVM_MEDIUM ) ;
128     endfunction
129 endclass
130 endpackage

```

## 6.8. FIFO coverage

```

1  package FIFO_coverage_pkg ;
2
3      import FIFO_seq_item_pkg::* ;
4      import uvm_pkg::* ;
5      `include "uvm_macros.svh"
6
7
8      class FIFO_coverage extends uvm_component;
9          `uvm_component_utils(FIFO_coverage)
10
11          uvm_analysis_export #(FIFO_seq_item) cov_export ;
12          uvm_tlm_analysis_fifo #(FIFO_seq_item) cov_fifo ;
13          FIFO_seq_item seq_item_cov ;
14
15          covergroup cvr_grp ;
16              wr_en_cp : coverpoint seq_item_cov.wr_en ;
17              rd_en_cp : coverpoint seq_item_cov.rd_en ;
18              wr_ack_cp : coverpoint seq_item_cov.wr_ack ;
19              overflow_cp : coverpoint seq_item_cov.overflow ;
20              full_cp : coverpoint seq_item_cov.full ;
21              empty_cp : coverpoint seq_item_cov.empty ;
22              almostfull_cp : coverpoint seq_item_cov.almostfull ;
23              almostempty_cp : coverpoint seq_item_cov.almostempty ;
24              underflow_cp : coverpoint seq_item_cov.underflow ;

```

```

27 wr_full_cp : cross wr_en_cp , full_cp ;
28 wr_wr_ack_cp : cross wr_en_cp , wr_ack_cp ;
29 wr_overflow_cp : cross wr_en_cp , overflow_cp ;
30 wr_empty_cp : cross wr_en_cp , empty_cp ;
31 wr_almostfull_cp : cross wr_en_cp , almostfull_cp ;
32 wr_almostempty_cp : cross wr_en_cp , almostempty_cp ;
33 wr_underflow_cp : cross wr_en_cp , underflow_cp ;
34
35 rd_full_cp : cross rd_en_cp , full_cp
36 {
37   ignore_bins rd_full = binsof(rd_en_cp) intersect {1} && binsof(full_cp) intersect {1} ;
38 }
39 rd_wr_ack_cp : cross rd_en_cp , wr_ack_cp ;
40 rd_overflow_cp : cross rd_en_cp , overflow_cp ;
41 rd_empty_cp : cross rd_en_cp , empty_cp ;
42 rd_almostfull_cp : cross rd_en_cp , almostfull_cp ;
43 rd_almostempty_cp : cross rd_en_cp , almostempty_cp ;
44 rd_underflow_cp : cross rd_en_cp , underflow_cp ;
45 endgroup
46
47 function new(string name = "FIFO_coverage" , uvm_component parent = null ) ;
48   super.new(name ,parent ) ;
49   cvr_grp=new();
50 endfunction
51
52 function void build_phase(uvm_phase phase) ;
53   super.build_phase(phase);
54   cov_export = new("cov_export" , this) ;
55   cov_fifo = new("cov_fifo" , this) ;
56 endfunction
57
58 function void connect_phase(uvm_phase phase) ;
59   super.connect_phase(phase);
60   cov_export.connect(cov_fifo.analysis_export);
61 endfunction
62
63 task run_phase(uvm_phase phase) ;
64   super.run_phase(phase);
65   forever
66   begin
67     cov_fifo.get(seq_item_cov) ;
68     cvr_grp.sample();
69   end
70 endtask
71 endclass
72 endpackage

```

## 6.9. FIFO Agent

```
1  package FIFO_agent_pkg ;
2
3      import FIFO_driver_pkg::*;
4      import FIFO_sequencer_pkg::*;
5      import FIFO_monitor_pkg::*;
6      import FIFO_config_pkg::*;
7      import FIFO_seq_item_pkg::* ;
8      import uvm_pkg::* ;
9      `include "uvm_macros.svh"
10
11
12  class FIFO_agent extends uvm_agent;
13      `uvm_component_utils(FIFO_agent)
14
15      FIFO_driver  driver ;
16      FIFO_sequencer sqr ;
17      FIFO_monitor mon ;
18      FIFO_config  FIFO_cfg ;
19      uvm_analysis_port #(FIFO_seq_item) agt_ap ;
20
21      function new(string name = "FIFO_agent" , uvm_component parent = null );
22          | super.new(name ,parent ) ;
23      endfunction
24
25      function void build_phase(uvm_phase phase) ;
26          super.build_phase(phase);
27          if(!uvm_config_db#(FIFO_config)::get (this , "" , "CFG" , FIFO_cfg ) )
28              `uvm_fatal("build_phase" , "driver - unable to get the virtual interface ") ;
29          driver = FIFO_driver::type_id::create("driver",this );
30          sqr = FIFO_sequencer::type_id::create("sqr",this );
31          mon = FIFO_monitor::type_id::create("mon",this );
32          agt_ap = new("agt_ap" , this) ;
33      endfunction
34
35      function void connect_phase(uvm_phase phase) ;
36          driver.FIFO_vif = FIFO_cfg.FIFO_vif ;
37          mon.FIFO_vif = FIFO_cfg.FIFO_vif ;
38          driver.seq_item_port.connect(sqr.seq_item_export) ;
39          mon.mon_ap.connect(agt_ap);
40      endfunction
41  endclass
42  endpackage
```

## 6.10. Driver

```
1 package FIFO_driver_pkg ;
2
3 import uvm_pkg::* ;
4 import FIFO_config_pkg::*;
5 import FIFO_seq_item_pkg::* ;
6 `include "uvm_macros.svh"
7
8
9 class FIFO_driver extends uvm_driver #(FIFO_seq_item);
10   `uvm_component_utils(FIFO_driver)
11   virtual FIFO_if FIFO_vif ;
12   FIFO_seq_item seq_item ;
13   function new(string name = "FIFO_driver" , uvm_component parent = null );
14     super.new(name ,parent ) ;
15   endfunction
16
17
18   task run_phase(uvm_phase phase) ;
19     super.run_phase(phase);
20     forever
21     begin
22       seq_item = FIFO_seq_item::type_id::create("seq_item");
23       seq_item_port.get_next_item(seq_item);
24       FIFO_vif.rst_n = seq_item.rst_n ;
25       FIFO_vif.wr_en = seq_item.wr_en ;
26       FIFO_vif.rd_en = seq_item.rd_en ;
27       FIFO_vif.data_in = seq_item.data_in ;
28       @(negedge FIFO_vif.clk ) ;
29       seq_item_port.item_done();
30       `uvm_info("run_phase" ,seq_item.convert2string_stimulus() , UVM_HIGH )
31     end
32   endtask
33 endclass
34 endpackage
```

## 6.11. Monitor

```
1  package FIFO_monitor_pkg ;
2
3  import uvm_pkg::* ;
4  import FIFO_seq_item_pkg::* ;
5  `include "uvm_macros.svh"
6
7
8  class FIFO_monitor extends uvm_monitor ;
9      `uvm_component_utils(FIFO_monitor)
10     virtual FIFO_if FIFO_vif ;
11     FIFO_seq_item  rsp_seq_item ;
12     uvm_analysis_port #(FIFO_seq_item) mon_ap ;
13
14     function new(string name = "FIFO_monitor" , uvm_component parent = null );
15         | super.new(name ,parent ) ;
16     endfunction
17
18     function void build_phase(uvm_phase phase) ;
19         | super.build_phase(phase);
20         | mon_ap = new("mon_ap" , this) ;
21     endfunction
22
23
24     task run_phase(uvm_phase phase) ;
25         | super.run_phase(phase);
26         | forever
27         | begin
28             | rsp_seq_item = FIFO_seq_item::type_id::create("rsp_seq_item");
29             | @(negedge FIFO_vif.clk ) ;
30             | rsp_seq_item.rst_n = FIFO_vif.rst_n ;
31             | rsp_seq_item.wr_en = FIFO_vif.wr_en ;
32             | rsp_seq_item.rd_en = FIFO_vif.rd_en ;
33             | rsp_seq_item.data_in = FIFO_vif.data_in ;
34
35             | rsp_seq_item.data_out = FIFO_vif.data_out ;
36             | rsp_seq_item.wr_ack = FIFO_vif.wr_ack ;
37             | rsp_seq_item.overflow = FIFO_vif.overflow ;
38             | rsp_seq_item.full = FIFO_vif.full ;
39             | rsp_seq_item.empty = FIFO_vif.empty ;
40             | rsp_seq_item.almostfull = FIFO_vif.almostfull ;
41             | rsp_seq_item.almostempty = FIFO_vif.almostempty ;
42             | rsp_seq_item.underflow = FIFO_vif.underflow ;
43
44             | mon_ap.write(rsp_seq_item);
45             | `uvm_info("run_phase" ,rsp_seq_item.convert2string_stimulus() , UVM_HIGH )
46
47         | end
48     endtask
49 endclass
50 endpackage
```

## 6.12. FIFO sequence item

```
1  package FIFO_seq_item_pkg ;
2
3  import uvm_pkg::* ;
4  `include "uvm_macros.svh"
5
6  class FIFO_seq_item extends uvm_sequence_item;
7      `uvm_object_utils(FIFO_seq_item)
8      parameter FIFO_WIDTH = 16;
9      parameter FIFO_DEPTH = 8;
10
11      rand bit rst_n;
12      rand bit wr_en;
13      rand bit rd_en ;
14      rand bit [FIFO_WIDTH-1:0] data_in;
15      bit [FIFO_WIDTH-1:0] data_out;
16      bit wr_ack, overflow;
17      bit full, empty, almostfull, almostempty, underflow;
18
19      constraint write_only{
20          wr_en dist {1:/100 , 0:/0 } ;
21          rd_en dist {1:/0 , 0:/100 } ;
22          rst_n dist {1:/99 , 0:/1 } ;
23      }
24      constraint read_only{
25          rd_en dist {1:/100 , 0:/0 } ;
26          wr_en dist {1:/0 , 0:/100 } ;
27          rst_n dist {1:/99 , 0:/1 } ;
28      }
29      constraint read_and_write{
30          wr_en dist {1:/70 , 0:/30 } ;
31          rd_en dist {1:/30 , 0:/70 } ;
32          rst_n dist {1:/99 , 0:/1 } ;
33      }
34
35      function new(string name = "FIFO_seq_item" );
36          super.new(name );
37      endfunction
38
39      function string convert2string();
40          return $sformatf ("%s , rst_n = %0d , wr_en = %0d , rd_en = %0d , data_out = %0d , wr_ack = %0d , data_in = %0d , overflow = %0d ,\
41              full = %0d , empty = %0d , almostfull = %0d , almostempty = %0d , underflow = %0d " , super.convert2string() , rst_n , wr_en , rd_en \
42              , data_out , wr_ack , data_in , overflow , full , empty , almostfull , almostempty , underflow );
43      endfunction
44
45      function string convert2string_stimulus();
46          return $sformatf ("rst_n = %0d , wr_en = %0d , rd_en = %0d , data_in = %0d" , rst_n , wr_en , rd_en , data_in );
47      endfunction
48  endclass
49  endpackage
```

## 6.13. FIFO sequencer

```
1 package FIFO_sequencer_pkg ;
2
3 import uvm_pkg::* ;
4 import FIFO_seq_item_pkg::* ;
5 `include "uvm_macros.svh"
6
7
8 class FIFO_sequencer extends uvm_sequencer #(FIFO_seq_item);
9     `uvm_component_utils(FIFO_sequencer)
10     function new(string name = "FIFO_sequencer" , uvm_component parent = null );
11         super.new(name ,parent ) ;
12     endfunction
13
14 endclass
15 endpackage
```

## 7. Do file


```
1 vlib work
2 vlog *v +cover
3 vsim -voptargs=+acc work.FIFO_top -classdebug -uvmcontrol=all -cover
4 add wave /FIFO_top/FIFOif/*
5 coverage save top.ucdb -onexit
6 run -all
7
8 quit -sim
9 vcover report top.ucdb -details -all -output coverage.txt
10
```

## 8. Transcript


```
# UVM_INFO FIFO_test.sv(44) @ 2: uvm_test_top [run_phase] reset deassert
# UVM_INFO FIFO_test.sv(46) @ 2: uvm_test_top [run_phase] write stimulus generation started
# UVM_INFO FIFO_test.sv(48) @ 202: uvm_test_top [run_phase] write stimulus generation ended
# UVM_INFO FIFO_test.sv(50) @ 202: uvm_test_top [run_phase] read stimulus generation started
# UVM_INFO FIFO_test.sv(52) @ 402: uvm_test_top [run_phase] read stimulus generation ended
# UVM_INFO FIFO_test.sv(55) @ 402: uvm_test_top [run_phase] read and write stimulus generation started
# UVM_INFO FIFO_test.sv(57) @ 20402: uvm_test_top [run_phase] read and write stimulus generation ended
# UVM_INFO verilog_src/uvm-1.1d/src/base/uvm_objection.svh(1268) @ 20402: reporter [TEST_DONE] 'run' phase is ready to proceed to the 'extract' phase
# UVM_INFO FIFO_scoreboard.sv(127) @ 20402: uvm_test_top.env.sb [report_phase] total successful 10201
# UVM_INFO FIFO_scoreboard.sv(128) @ 20402: uvm_test_top.env.sb [report_phase] total FAILED 0
#
# --- UVM Report Summary ---
#
# ** Report counts by severity
# UVM_INFO : 14
# UVM_WARNING : 0
# UVM_ERROR : 0
# UVM_FATAL : 0
# ** Report counts by id
# [Questa UVM] 2
# [RNTST] 1
# [TEST_DONE] 1
# [report_phase] 2
# [run_phase] 8
# ** Note: $finish : C:/questasim64_10.4c/win64/./verilog_src/uvm-1.1d/src/base/uvm_root.svh(430)
```




## 9. Waveform write sequence



## 10. Waveform read sequence



## 11. Waveform write and read sequence



## 12. Assertion

	File	Type	Mode	Priority	Period	Phase	Offset	Time	Unit	Value	Condition	Action
	/FIFO_seq_main_p...	Immediate	SVA	on	0	1	-	-	-	-	off	assert (randomize(...))
	/FIFO_seq_main_p...	Immediate	SVA	on	0	1	-	-	-	-	off	assert (randomize(...))
	/FIFO_seq_main_p...	Immediate	SVA	on	0	1	-	-	-	-	off	assert (randomize(...))
(+)	/FIFO_top/dU/sva/...	Concurrent	SVA	on	0	1	-	0B	0B	0 ns	0 off	assert( @posedge FIFOif.clk) disable iff (~FIFOif.rs
(+)	/FIFO_top/dU/sva/...	Concurrent	SVA	on	0	1	-	0B	0B	0 ns	0 off	assert( @posedge FIFOif.clk) disable iff (~FIFOif.rs
	/FIFO_top/dU/sva/...	Immediate	SVA	on	0	1	-	-	-	-	off	assert (FIFOif.empty&~FIFOif.full&~FIFOif.a...)
	/FIFO_top/dU/sva/...	Immediate	SVA	on	0	1	-	-	-	-	off	assert ((FIFOif.empty~ FIFOif.full)&FIFOif....)
	/FIFO_top/dU/sva/...	Immediate	SVA	on	0	1	-	-	-	-	off	assert ((FIFOif.empty~ FIFOif.full)&~FIFOif....)
	/FIFO_top/dU/sva/...	Immediate	SVA	on	0	1	-	-	-	-	off	assert (~FIFOif.empty&FIFOif.full&~FIFOif.a...)
(+)	/FIFO_top/dU/sva/...	Concurrent	SVA	on	0	1	-	0B	0B	0 ns	0 off	assert( @posedge FIFOif.clk) disable iff (~FIFOif.rs
(+)	/FIFO_top/dU/sva/...	Concurrent	SVA	on	0	1	-	0B	0B	0 ns	0 off	assert( @posedge FIFOif.clk) disable iff (~FIFOif.rs
(+)	/FIFO_top/dU/sva/...	Concurrent	SVA	on	0	1	-	0B	0B	0 ns	0 off	assert( @posedge FIFOif.clk) disable iff (~FIFOif.rs
(+)	/FIFO_top/dU/sva/...	Concurrent	SVA	on	0	1	-	0B	0B	0 ns	0 off	assert( @posedge FIFOif.clk) disable iff (~FIFOif.rs
(+)	/FIFO_top/dU/sva/...	Concurrent	SVA	on	0	1	-	0B	0B	0 ns	0 off	assert( @posedge FIFOif.clk) disable iff (~FIFOif.rs

### 13. Assertion coverage

Item	Language	Kind	Size	Count	Process	Time	Height	Simple %	Simple Graph	Analysis	Memory	Test Memory	Test Memory	Time	Comments	Process
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	3717	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	3194	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	151	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	179	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	1734	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	1460	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	151	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	2595	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	846	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	43	1	Unli...	1	100%		✓	0	0	0	ns	0		
🔗 /FIFO_top/dU/sva/... SVA	✓	Off	956	1	Unli...	1	100%		✓	0	0	0	ns	0		

## 14. Code Coverage

```

3  =====
4  == File: FIFO.sv
5  =====
6  Statement Coverage:
7  Enabled Coverage      Active      Hits      Misses % Covered
8  -----
9  Stmt's                25          25          0      100.0
10
11 =====Statement Details=====
12
13 Statement Coverage for file FIFO.sv --
14
15 1 //////////////////////////////////////
16 2 // Author: Kareem Waseem
17 3 // Course: Digital Verification using SV & UVM
18 4 //
19 5 // Description: FIFO Design
20 6 //
21 7 //////////////////////////////////////
22 8 module FIFO(FIFO_inter.DUT FIFO_if);
23 9
24 10 localparam max_fifo_addr = $clog2(FIFO_if.FIFO_DEPTH);
25 11
26 12 reg [FIFO_if.FIFO_WIDTH-1:0] mem [FIFO_if.FIFO_DEPTH-1:0];
27 13
28 14 reg [max_fifo_addr-1:0] wr_ptr, rd_ptr;
29 15 reg [max_fifo_addr:0] count;
30 16
31 17 1 101004 always @(posedge FIFO_if.clk or negedge FIFO_if.rst_n) begin
32 18 18 if (!FIFO_if.rst_n) begin
33 19 19 1 2011 wr_ptr <= 0;
34 20 20 end
35 21 21 else if (FIFO_if.wr_en && count < FIFO_if.FIFO_DEPTH) begin

```

```

183
184 Branch Coverage:
185   Enabled Coverage      Active      Hits      Misses % Covered
186   -----
187   Branches              23        23         0    100.0
188
189 =====Branch Details=====
190
191 Branch Coverage for file FIFO.sv --
192
193 -----IF Branch-----
194   18                      101004    Count coming in to IF
195   18                      2011      if (!FIFO_if.rst_n) begin
196   21                      35936    else if (FIFO_if.wr_en && count < FIFO_if.FIFO_DEPTH) begin
197   26                      63057    else begin
198 Branch totals: 3 hits of 3 branches = 100.0%
199
200 -----IF Branch-----
201   28                      63057    Count coming in to IF
202   28                      33332    if (FIFO_if.full && FIFO_if.wr_en)
203   30                      29725    else
204 Branch totals: 2 hits of 2 branches = 100.0%
205
206 -----IF Branch-----
207   36                      101004    Count coming in to IF
208   36                      2011      if (!FIFO_if.rst_n) begin
209   40                      29029    else if (FIFO_if.rd_en && count != 0) begin
210   45                      516        if (FIFO_if.empty && FIFO_if.rd_en)
211   47                      69448    else
212 Branch totals: 4 hits of 4 branches = 100.0%
401
402 =====Toggle Details=====
403
404 Toggle Coverage for File FIFO.sv --
405
406   |      Line      |      Node      |      1H->0L      |      0L->1H      |      "Coverage"
407   |-----|
408   |      14      | wr_ptr[2]      |      1      |      1      |      100.00
409   |      14      | wr_ptr[1]      |      1      |      1      |      100.00
410   |      14      | wr_ptr[0]      |      1      |      1      |      100.00
411   |      14      | rd_ptr[2]      |      1      |      1      |      100.00
412   |      14      | rd_ptr[1]      |      1      |      1      |      100.00
413   |      14      | rd_ptr[0]      |      1      |      1      |      100.00
414   |      15      | count[3]       |      1      |      1      |      100.00
415   |      15      | count[2]       |      1      |      1      |      100.00
416   |      15      | count[1]       |      1      |      1      |      100.00
417   |      15      | count[0]       |      1      |      1      |      100.00
418
419 Total Node Count      =      10
420 Toggled Node Count    =      10
421 Untoggled Node Count  =      0
422
423 Toggle Coverage      =      100.0% (20 of 20 bins)
424
425

```

## 15. Functional Coverage report

Covergroup	Metric	Goal	Status
TYPE /FIFO_coverage_pkg/FIFO_coverage/cvr_grp	100.0%	100	Covered
covered/total bins:	73	73	
missing/total bins:	0	73	
% Hit:	100.0%	100	
Coverpoint cvr_grp::wr_en_cp	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	3105	1	Covered
bin auto[1]	7096	1	Covered
Coverpoint cvr_grp::rd_en_cp	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	7089	1	Covered
bin auto[1]	3112	1	Covered
Coverpoint cvr_grp::wr_ack_cp	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	6378	1	Covered
bin auto[1]	3823	1	Covered
Coverpoint cvr_grp::overflow_cp	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	5959	1	Covered
bin auto[1]	4242	1	Covered
Coverpoint cvr_grp::full_cp	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	5470	1	Covered
bin auto[1]	4731	1	Covered
Coverpoint cvr_grp::empty_cp	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	

<u>Coverpoint cvr_grp::empty_cp</u>	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	9895	1	Covered
bin auto[1]	306	1	Covered
<u>Coverpoint cvr_grp::almostfull_cp</u>	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	7295	1	Covered
bin auto[1]	2906	1	Covered
<u>Coverpoint cvr_grp::almostempty_cp</u>	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	9888	1	Covered
bin auto[1]	313	1	Covered
<u>Coverpoint cvr_grp::underflow_cp</u>	100.0%	100	Covered
covered/total bins:	2	2	
missing/total bins:	0	2	
% Hit:	100.0%	100	
bin auto[0]	10025	1	Covered
bin auto[1]	176	1	Covered
<u>Cross cvr_grp::wr_full_cp</u>	100.0%	100	Covered
covered/total bins:	4	4	
missing/total bins:	0	4	
% Hit:	100.0%	100	
bin <auto[0],auto[0]>	2097	1	Covered
bin <auto[1],auto[0]>	3373	1	Covered
bin <auto[0],auto[1]>	1008	1	Covered
bin <auto[1],auto[1]>	3723	1	Covered
<u>Cross cvr_grp::wr_wr_ack_cp</u>	100.0%	100	Covered
covered/total bins:	4	4	
missing/total bins:	0	4	
% Hit:	100.0%	100	
bin <auto[0],auto[0]>	3089	1	Covered
bin <auto[1],auto[0]>	3289	1	Covered
bin <auto[0],auto[1]>	16	1	Covered
bin <auto[1],auto[1]>	3807	1	Covered